# Maintainer: kpcyrd <kpcyrd[at]archlinux[dot]org>
# Maintainer: Caleb Maclennan <caleb@alerque.com>
# Contributor: Joaquim Monteiro <joaquim.monteiro@protonmail.com>

_pyname=iOSbackup
pkgname=python-iosbackup
pkgver=0.9.925
pkgrel=2
pkgdesc='Library for reading and extracting files from password-encrypted iOS backups'
arch=(any)
url='https://github.com/avibrazil/iOSbackup'
license=('LGPL')
depends=('python' 'python-nskeyedunarchiver' 'python-pycryptodome')
makedepends=(python-{build,installer,wheel} 'python-setuptools')
source=("https://files.pythonhosted.org/packages/source/${_pyname::1}/${_pyname}/${_pyname}-${pkgver}.tar.gz")
sha512sums=('b8339359f806f9629736a7abe9e35a6b868c08529b43d0b9e0408afca25babc167ff70abb2516f557ac071fa26193d1cfaf1348f7bfee0eb57ef736c1c5d83d0')

build() {
    cd "${_pyname}-${pkgver}"
    python -m build -wn
}

package() {
    cd "${_pyname}-${pkgver}"
    python -m installer -d "$pkgdir" dist/*.whl
}

# vim: ts=2 sw=2 et:
